

```

1      * BSTTEZA.C  COPYRIGHT BSI 2001-2009
2      *      PLACE THE FOLLOWING LINE IN THE WORKING STORAGE SECTION.
3      *01  EZA-CALL-DATA.          COPY BSTTEZA.
4      *
5          05  EZA-FUNCTION          PIC X(16) .
6          05  EZA-AF                PIC S9(09) COMP VALUE +2.
7          05  EZA-BACKLOG           PIC S9(09) COMP VALUE +10.
8          05  EZA-CANNLEN           PIC S9(09) COMP.
9          05  EZA-CLIENT.
10         10  EZA-CLIENT-DOMAIN     PIC S9(09) COMP.
11         10  EZA-CLIENT-NAME       PIC X(08) .
12         10  EZA-CLIENT-TASK       PIC X(08) .
13         10  EZA-CLIENT-TOKEN      PIC X(04) .
14         10  FILLER                PIC X(16) .
15        05  EZA-COMMAND-X.
16         10  EZA-COMMAND           PIC S9(09) COMP.
17        05  EZA-DSTADDR            PIC X(45) .
18        05  EZA-DSTLEN             PIC S9(04) COMP.
19        05  EZA-DOMAIN             PIC S9(09) COMP.
20        05  EZA-ERRNO             PIC S9(09) COMP.
21        05  EZA-FLAGS             PIC S9(09) COMP.
22        05  EZA-HINTS              POINTER.
23        05  EZA-HOSTADDR           PIC X(04) .
24        05  EZA-HOSTENT            POINTER.
25        05  EZA-HOW               PIC S9(09) COMP.
26        05  EZA-IDENT              VALUE SPACES.
27         10  EZA-IDENT-TCPNAME.
28         15  EZA-IDENT-TAG         PIC X(06) .
29         15  EZA-IDENT-SYSID      PIC X(02) .
30         10  EZA-IDENT-ADSNAME     PIC X(08) .
31        05  EZA-MAXSNO             PIC S9(09) COMP VALUE +0.
32        05  EZA-MAXSOC            PIC S9(04) COMP.
33        05  EZA-MAXSOC-SELECT      PIC S9(09) COMP.
34        05  EZA-NAMEV6.
35         10  EZA-NAMEV6-FAMILY-X.
36         15  EZA-NAMEV6-FAMILY     PIC S9(04) COMP.
37         10  EZA-NAMEV6-PORTX.
38         15  EZA-NAMEV6-PORT      PIC 9(04) COMP.
39         10  EZA-NAMEV6-FLOW       PIC X(04) .
40         10  EZA-NAMEV6-IPADDRESS  PIC X(16) .
41         10  EZA-NAMEV6-SCOPE     PIC X(04) .
42        05  EZA-NAME              REDEFINES EZA-NAMEV6.
43         10  EZA-NAME-FAMILY-X.
44         15  EZA-NAME-FAMILY     PIC S9(04) COMP.
45         10  EZA-NAME-PORTX.
46         15  EZA-NAME-PORT      PIC 9(04) COMP.
47         10  EZA-NAME-IPADDRESS  PIC X(04) .
48         10  FILLER                PIC X(08) .
49         10  FILLER-TO-MATCH-V6   PIC X(12) .
50        05  EZA-NAMELEN           PIC S9(09) COMP.
51        05  EZA-NODELEN           PIC S9(09) COMP.
52        05  EZA-NTOP-FAMILY       PIC S9(09) COMP.
53        05  EZA-NBYTE            PIC S9(09) COMP.
54        05  EZA-OPTLEN           PIC S9(09) COMP.

```

```

55      05  EZA-OPTNAME          PIC X(04) .
56      05  EZA-OPTVAL          PIC X(16) .
57      05  EZA-PROTO          PIC S9(09) COMP .
58      05  EZA-RES-X .
59          10  EZA-RES          POINTER .
60      05  EZA-RETCODE-X .
61          10  EZA-RETCODE      PIC S9(09) COMP .
62      05  EZA-S-X .
63          10  EZA-S            PIC S9(04) COMP .
64      05  EZA-S-ACCEPT        PIC S9(04) COMP .
65      05  EZA-SERVICE        PIC X(32)      VALUE SPACES .
66      05  EZA-SERVLN         PIC S9(09) COMP .
67      05  EZA-SOCRECV        PIC S9(04) COMP .
68      05  EZA-SOCTYPE        PIC S9(09) COMP VALUE +1 .
69          88  EZA-SOCTYPE-STREAM          VALUE +1 .
70          88  EZA-SOCTYPE-DATAGRAM        VALUE +2 .
71          88  EZA-SOCTYPE-RAW            VALUE +3 .
72      05  EZA-SUBTASK          VALUE SPACES .
73          10  EZA-SUBTASK-9X7          PIC 9(07) .
74          10  EZA-SUBTASK-TYPE        PIC X(01) .
75      05  EZA-TIMEOUT .
76          10  EZA-TIMEOUT-SECONDS     PIC S9(09) COMP .
77          10  EZA-TIMEOUT-MICROS      PIC S9(09) COMP .
78      *
79      05  EZA-ADDRINFO .
80          10  EZA-AI-FLAGS          PIC S9(09) COMP .
81          10  EZA-AI-FAMILY        PIC S9(09) COMP .
82          10  EZA-AI-SOCKTYPE      PIC S9(09) COMP .
83          10  EZA-AI-PROTOCOL      PIC S9(09) COMP .
84          10  EZA-AI-ADDRLEN       PIC S9(09) COMP .
85          10  EZA-AI-CANON-PTR     POINTER .
86          10  EZA-AI-ADDR-PTR     POINTER .
87          10  EZA-AI-NEXT-PTR     POINTER .
88      *
89      *      THE FOLLOWING ARE CONSTANTS USED BY SOME OPERATIONS.
90      *
91      05  EZA-INETANY          PIC S9(04) COMP VALUE +0 .
92      05  EZA-INET            PIC S9(04) COMP VALUE +2 .
93      05  EZA-INET6          PIC S9(04) COMP VALUE +19 .
94      05  EZA-STREAM          PIC S9(04) COMP VALUE +1 .
95      05  EZA-DATAGRAM        PIC S9(04) COMP VALUE +2 .
96      05  EZA-RAW            PIC S9(04) COMP VALUE +3 .
97      05  EZA-TCP            PIC S9(04) COMP VALUE +6 .
98      05  EZA-UDP            PIC S9(04) COMP VALUE +17 .
99      05  EZA-AI-PASSIVE      PIC S9(04) COMP VALUE +1 .
100     05  EZA-AI-CANONNAMEOK   PIC S9(04) COMP VALUE +2 .
101     05  EZA-AI-NUMERICHOST   PIC S9(04) COMP VALUE +4 .
102     05  EZA-AI-NUMERICSERV   PIC S9(04) COMP VALUE +8 .
103     05  EZA-AI-V4MAPPED      PIC S9(04) COMP VALUE +16 .
104     05  EZA-AI-ALL           PIC S9(04) COMP VALUE +32 .
105     05  EZA-AI-ADDRCONFIG    PIC S9(04) COMP VALUE +64 .

```

```

106
107 Excerpt from BSTTEZAP.C
108     EZA-GETADDRINFO SECTION.
109     GETADDRINFO-START.
110     * BEFORE PERFORMING THIS SECTION, SOME FIELDS NEED TO BE SET.
111     * THE FOLLOWING 8 LINES ARE EXAMPLES OF SETTING REQUIRED FIELDS.
112     *     MOVE <HOST NAME> TO BUFFER-OUT.
113     *     MOVE LENGTH OF <HOST NAME> TO EZA-NODELEN.
114     *     MOVE <SERVICE> OR SPACES TO EZA-SERVICE.
115     *     MOVE LENGTH OF <SERVICE> OR ZEROES TO EZA-SERVLEN.
116     *     MOVE LOW-VALUES TO EZA-ADDRINFO.
117     *     MOVE <EZA-INET | EZA-INET6 | EZA-INETANY> TO EZA-AI-FAMILY.
118     *     COMPUTE EZA-AI-FLAGS = <ADD TOGETHER EZA-AI- FLAG VALUES>
119     *     MOVE EZA-STREAM TO EZA-AI-SOCKTYPE.
120     MOVE 'GETADDRINFO' TO EZA-FUNCTION.
121     PERFORM CONSOLE-TRACE.
122     MOVE +0 TO EZA-AI-PROTOCOL.
123     MOVE +0 TO EZA-AI-ADDRLEN.
124     SET EZA-AI-CANON-PTR TO NULLS.
125     SET EZA-AI-ADDR-PTR TO NULLS.
126     SET EZA-AI-NEXT-PTR TO NULLS.
127     CALL 'BSTTEZSP' USING EZA-ADDRINFO EZA-HINTS.
128     MOVE +0 TO EZA-CANNLEN.
129     MOVE +0 TO EZA-ERRNO.
130     MOVE +0 TO EZA-RETCODE.
131     CALL 'EZASOKET'
132         USING
133         EZA-FUNCTION
134         BUFFER-OUT
135         EZA-NODELEN
136         EZA-SERVICE
137         EZA-SERVLEN
138         EZA-HINTS
139         EZA-RES
140         EZA-CANNLEN
141         EZA-ERRNO
142         EZA-RETCODE.
143     MOVE EZA-FUNCTION TO CURRENT-FUNCTION.
144     MOVE EZA-ERRNO TO CURRENT-ERROR.
145     GETADDRINFO-EXIT.
146     EXIT.
147 ...
148     EZA-FREEADDRINFO SECTION.
149     FREEADDRINFO-START.
150     MOVE 'FREEADDRINFO' TO EZA-FUNCTION.
151     PERFORM CONSOLE-TRACE.
152     MOVE +0 TO EZA-ERRNO.
153     MOVE +0 TO EZA-RETCODE.
154     CALL 'EZASOKET'
155         USING
156         EZA-FUNCTION
157         EZA-RES
158         EZA-ERRNO
159         EZA-RETCODE.

```

```

160         MOVE EZA-FUNCTION TO CURRENT-FUNCTION.
161         MOVE EZA-ERRNO TO CURRENT-ERROR.
162     FREEADDRINFO-EXIT.
163         EXIT.
164 ...
165     EZA-RES-EXTRACT SECTION.
166     ERE-START.
167     * BEFORE PERFORMING THIS SECTION, SOME FIELDS NEED TO BE SET.
168     * THE FOLLOWING 1 LINE IS AN EXAMPLE OF SETTING REQUIRED FIELD
169     * BUT *NOTE* ONLY FOR THE FIRST CALL TO THIS ROUTINE AFTER
170     * PERFORMING EZA-GETADDRINFO. TO EXTRACT ADDITIONAL ADDRESSES,
171     * EZ9-RES MUST BE SET TO THE VALUE OF EZ9-RES-NEXT. ALL ADDRESSES
172     * HAVE BEEN EXTRACTED ONCE EZ9-RES-NEXT-X IS EQUAL TO LOW-VALUES.
173     *     SET EZ9-RES TO EZA-RES.
174         MOVE 'EZACIC09' TO CURRENT-FUNCTION.
175         IF EZ9-RES-X IS EQUAL TO LOW-VALUES
176             MOVE -2 TO EZA-RETCODE
177             GO TO ERE-EXIT.
178         MOVE ZEROS TO EZ9-NAME-LEN.
179         MOVE SPACES TO EZ9-CANONICAL-NAME.
180         SET EZ9-NAME TO NULLS.
181         SET EZ9-RES-NEXT TO NULLS.
182         MOVE +0 TO EZA-RETCODE.
183         CALL 'EZACIC09'
184             USING
185             EZ9-RES
186             EZ9-NAME-LEN
187             EZ9-CANONICAL-NAME
188             EZ9-NAME
189             EZ9-RES-NEXT
190             EZA-RETCODE.
191         IF EZA-RETCODE IS LESS THAN +0
192             GO TO ERE-EXIT.
193     *     SET ADDRESS OF <OUTPUT-IP-NAME> TO EZ9-NAME.
194     *     NOTE: <OUTPUT-IP-NAME> MUST BE IN THE LINKAGE SECTION AND
195     *     MUST HAVE A LENGTH OF AT LEAST 28 BYTES.
196     *     MOVE <OUTPUT-IP-NAME> (1: EZ9-NAME-LEN) TO EZA-NAME.
197     *     SET EZ9-RES TO EZ9-RES-NEXT.
198     ERE-EXIT.
199         EXIT.

```

```

200
201 Excerpt from BSTTEZ55.C Sample Client
202
203     ML-INITAPI.
204         MOVE +0 TO EZA-MAXSOC.
205         MOVE SPACES TO EZA-IDENT.
206         MOVE SPACES TO EZA-SUBTASK.
207         PERFORM EZA-INITAPI.
208         IF EZA-RETCODE IS LESS THAN +0
209             GO TO AR-0001.
210     ML-GETADDRINFO.
211         MOVE 'LOCALHOST' TO BUFFER-OUT.
212         MOVE +9 TO EZA-NODELEN.
213         MOVE SPACES TO EZA-SERVICE.
214         MOVE +0 TO EZA-SERVLEN.
215         MOVE LOW-VALUES TO EZA-ADDRINFO.
216         COMPUTE EZA-AI-FLAGS = EZA-AI-ALL.
217         MOVE EZA-INETANY TO EZA-AI-FAMILY.
218         MOVE EZA-STREAM TO EZA-AI-SOCKTYPE.
219         PERFORM EZA-GETADDRINFO.
220         IF EZA-RETCODE IS LESS THAN +0
221             GO TO AR-0008.
222     ML-EZACIC09.
223         SET EZ9-RES TO EZA-RES.
224         PERFORM EZA-RES-EXTRACT.
225         IF EZA-RETCODE IS LESS THAN +0
226             GO TO AR-0009.
227         SET ADDRESS OF OUTPUT-IP-NAME TO EZ9-NAME.
228         MOVE OUTPUT-IP-NAME (1:EZ9-NAME-LEN) TO EZA-NAME.
229     ML-FREEADDRINFO.
230         PERFORM EZA-FREEADDRINFO.
231         IF EZA-RETCODE IS LESS THAN +0
232             GO TO AR-0010.
233     ML-SOCKET.
234         MOVE EZA-NAME-FAMILY TO EZA-AF.
235         PERFORM EZA-SOCKET.
236         IF EZA-RETCODE IS LESS THAN +0
237             GO TO AR-0002.
238         MOVE EZA-RETCODE TO EZA-S.
239     ML-CONNECT.
240     *   MOVE LOW-VALUES TO EZA-NAME.
241     *   MOVE EZA-INET TO EZA-NAME-FAMILY.
242         MOVE WS-IPPORT TO EZA-NAME-PORT.
243     *   MOVE WS-IPADDRESS TO EZA-NAME-IPADDRESS.
244         PERFORM EZA-CONNECT.
245         IF EZA-RETCODE IS LESS THAN +0
246             GO TO AR-0003.

```